

COMS W4995 002: Parallel Functional Programming Fall 2021

PROJECT REPORT

 $Professor\ Stephen\ A.\ Edwards$

PowerList

Yash Datta

UNI yd2590



${\bf Contents}$

1	Intr	roducti	on																					2
2	Pro	ject D	escriptio	on																				2
	2.1	Usage								 				 			 					 		3
	2.2	Powerl	list in Ha	ask	ell					 				 			 					 		4
		2.2.1	Powerlis	st a	as Lis	st .				 				 			 					 		4
		2.2.2	Powerlis	st a	as Ur	nboxe	d Ve	ecto	r .	 				 			 					 		5
	2.3	Algori	thms							 				 			 					 		6
		2.3.1	Simple 1																					6
			2.3.1.1			L																		6
			2.3.1.2	S	SPSP	LPar	1			 				 								 		7
			2.3.1.3	-		LPar																		7
			2.3.1.4	-		LPar																		7
			2.3.1.5	-		BVec																		8
		2.3.2	Ladner	-																				8
		2.0.2	2.3.2.1			$\mathbf{e}_{\mathbf{ar}}$.																		9
			2.3.2.2			JBVe																		9
			2.3.2.3			Chunk				 		-		 	-		 -	-			-	 	 -	10
		2.3.3	Batcher							 				 					 -			 		11
		2.5.5	Dateller	L IVI	terge	5016				 		•	•	 	•	•	 •	•	 •		•	 •	 •	11
3	Ber	chmar	k																					13
	3.1	Setup								 				 								 		13
	3.2		S																					14
	0.2	3.2.1	Scan																					14
		3.2.1	Sort																					
		0.4.4	5016		• •					 	• •	•	•	 • •	•	•	 •	•	 •	• •	•	 •	 •	10
4	Cor	ıclusioı	and Fu	uti	ure '	Work	ζ																	19



1 Introduction

- J. Misra [1], has introduced **powerlist**, a new recursive data structure that permits concise and elegant description of many data parallel algorithms like Prefix-sum, Batcher's sorting schemes, FFT etc. Similar to a list structure, the base case of powerlist is a list of one element. Longer power lists are constructed from the elements of 2 powerlists, p and q, of the same length using 2 operators described below.
 - $p \mid q$ is the powerlist formed by concatenating p and q. This is called **tie**.
 - $p \bowtie q$ is the powerlist formed by successively taking alternate items from p and q, starting with p. This is called **zip**.

Further, both p and q are restricted to contain similar elements.

Following additional operators are necessary to express algorithms in terms of powerlists:

- $p \oplus q$ is the powerlist obtained by applying the binary scalar operator \oplus on the elements of p and q at the same position in the 2 lists.
- L^* is the powerlist obtained by shifting the powerlist L by one. The effect of shifting is to append a 0 to the left and discard the rightmost element.

Note that 0 is considered the left identity element of \oplus , i.e. $0 \oplus x = x$.

In the following examples, elements of powerlist are enclosed within angular brackets.

- $\langle 0 \rangle \mid \langle 1 \rangle = \langle 0 | 1 \rangle$
- $\langle 0 \rangle \bowtie \langle 1 \rangle = \langle 0 \ 1 \rangle$
- $\langle 0 \ 1 \rangle \mid \langle 2 \ 3 \rangle = \langle 0 \ 1 \ 2 \ 3 \rangle$
- $\langle 0 \ 1 \rangle \bowtie \langle 2 \ 3 \rangle = \langle 0 \ 2 \ 1 \ 3 \rangle$

There are many applications of this structure for parallel algorithms, some of which are:

- Prefix-sum (scan)
- Batcher sort
- Rank sort
- Fast Fourier Transform

In this project, I have developed a haskell module to demonstrate the use of powerlist in the following algorithms:

- scan: Several variations of scan algorithms have been developed, which are primarily of 2 types
 - Simple Prefix Sum
 - Ladner Fischer [1]
- sort: A Batcher merge sort scheme has been implemented for sorting using powerlist.

2 Project Description

This haskell project has been developed as a benchmark utility for running and comparing different parallel algorithms using powerlist. The project uses stack for building and creates 2 executables:

- powerlist-exe: for executing and analyzing each of the different algorithms individually
- powerlist-benchmark: for benchmarking each of the algorithm functions by executing them multiple times. This uses criterion package, hence all command line options of criterion can be used.

The project source code including detailed benchmarks are hosted on github here.



2.1 Usage

To run each of the algorithms, use the powerlist-exe executable, which supports 2 commands scan and sort:

```
> stack exec powerlist—exe — scan

Usage: powerlist—exe scan (-a|--algo ALGONAME) (-s|--size INPSIZE)

[-c|--csize CHUNKSIZE]

Run Scan Algorithm

Available options:

-a,--algo ALGONAME

Supported Algos:

[SPS,SPSPL,SPSPLPar1,SPSPLPar2,SPSPLPar3,LDF,LDFPar,SPSUBVecPLPar,LDFUBVecPLPar,LDFChunkUBVecPLPar]

-s,--size INPSIZE

Size of array in terms of powers of 2 on which to run scan

-c,--csize CHUNKSIZE

Size of chunks for parallelization
```

and

-h,--**help**

```
> stack exec powerlist—exe — sort 
 Usage: powerlist—exe sort (-a|--algo ALGONAME) (-s|--size INPSIZE)  [-c|--c size \ CHUNKSIZE]  Run Sort Algorithm
```

Show this help text

Available options:

```
-a,—algo ALGONAME Supported Algos: [DEFAULT,BATCHER]
-s,—size INPSIZE Size of array in terms of powers of 2 on which to run sort
-c,—csize CHUNKSIZE Size of chunks for parallelization
-h,—help Show this help text
```

For example to run simple prefix sum algorithm using powerlist on an array of input size 2^5 :

```
> stack exec powerlist-exe — scan —algo SPSPL —size 5 5984
```

To run the parallel version of the same algorithm using powerlist, on 8 cores and generate eventlog file for threadscope analysis:

```
> stack exec powerlist-exe — scan —algo SPSPLPar1 —size 20 +RTS -N8 -ls 192154133857304576
```

To run the benchmark for the same algorithm, use powerlist-benchmark executable:

```
> stack exec powerlist-bench — main/scan/par/nc/SPSPLPar1 +RTS -N8 benchmarking main/scan/par/nc/SPSPLPar1
```



Input Output

The input to **scan** algorithm is generated as a simple list/vector of length 2^d from 1 to d. The **scan** algorithm when run simply outputs the sum of the elements of the prefix sum array. So for example for input [1, 2, 3, 4], prefix sum array = [1, 3, 6, 10], and sum of the elements of this array = 20.

The input to **sort** is generated as a reverse list/vector of length 2^d from 2^d down to 1. The **sort** algorithm when run simply outputs the last element of the sorted array. This is done to prevent I/O from affecting algorithm performance in individual runs. Note that the benchmark utility benchmarks the functions directly, so is not affected by any IO.

Here d is the value of the supplied –size param.

2.2 Powerlist in Haskell

I have used 2 different implementations of powerlists, one using List and the other using Data. Vector. Unboxed

2.2.1 Powerlist as List

The List implementation of powerlist is straightforward and allows to implement the required operators easily:

```
-- Using simple list here as it would be most performant
type PowerList a = [a]
```

The tie function is same as ++ of List in haskell, but zip is a bit different which is shown below:

```
zip :: PowerList a -> PowerList a
{-# INLINE zip #-}
zip [] [] = []
zip xs ys = Prelude.zip xs ys >>= \(a, b) -> [a, b]
```

There is an analogous unzip function that is required for deconstructing the input powerlist into 2 smaller powerlists.

The L^* (shift) operator is implemented as below:

```
1 -- Right shift and use zero
```



```
rsh :: a -> PowerList a -> PowerList a
rsh zero xs = zero : init xs
```

2.2.2 Powerlist as Unboxed Vector

Unboxed Vectors are more memory friendly. We can further reduce memory usage by converting to mutable vectors and modifying the data in place, instead of creating more copies. But the implementation of operators in terms of Vectors is a bit more involved:

```
import qualified Data. Vector. Split as S
   import qualified Data.Vector.Unboxed as V
   import qualified Data.Vector.Unboxed.Mutable as M
   type PowerList a = V.Vector a
   tie :: V.Unbox a => PowerList a -> PowerList a -> PowerList a
   tie = (V.++)
   zip :: (V.Unbox a, Num a) => PowerList a -> PowerList a -> PowerList a
10
   {-# INLINE zip #-}
   zip xs ys =
12
13
     V.create $ do
       m <- M.new n
14
       write m 0
15
       return m
       n = V.length xs + V.length ys
18
       write m i
19
         | i < n = do
20
           M.unsafeWrite m i (xs V.! (i 'div' 2))
21
           M.unsafeWrite m (i + 1) (ys V.! (i 'div' 2))
22
           write m (i + 2)
         | otherwise = return ()
24
25
   zipWith ::
26
        (Num a, V.Unbox a)
27
     => (a -> a -> a)
28
     -> PowerList a
     -> PowerList a
     -> PowerList a
    {-# INLINE zipWith #-}
32
   zipWith op xs ys =
33
     V.create $ do
34
       m <- V.thaw xs
35
       write m ys 0
36
       return m
     where
38
       k = V.length xs
39
       write m y i
         | i < k = do
41
           curr <- M.unsafeRead m i</pre>
42
           M.unsafeWrite m i (op (y V.! i) curr)
           write m y (i + 1)
         | otherwise = return ()
45
```



```
unzip :: V.Unbox a => PowerList a -> (PowerList a, PowerList a)
unzip k = (b, c)
where
b = V.ifilter (\i _ -> even i) k
c = V.ifilter (\i _ -> odd i) k
```

The zipWith is nothing but \oplus operator described previously. Parallel versions of these operators have also been implemented, by splitting the input into chunks and running the operator in parallel on the chunks, and later combining the results.

2.3 Algorithms

As mentioned previously, many different versions of some algorithms, have been implemented with several optimizations to parallelize them effectively. The different algorithms and techniques used are described below.

2.3.1 Simple Prefix Sum

Prefix sum or scan is used to generate a prefix sum array from the input array, by summing all the elements of the array upto each element. So for example the prefix sum for input array [1, 2, 3, 4] is given by [1, 3, 6, 10](1 = 1, 3 = (1 + 2), 6 = (1 + 2 + 3), 10 = (1 + 2 + 3 + 4)) This is nothing but scanl1 in haskell.

```
Prelude> scanl1 (+) [1, 2, 3, 4] [1,3,6,10]
```

The following variations of this simple algorithm have been implemented:

- SPSPL: A sequential prefix sum using powerlist, to demonstrate equivalence.
- SPSPLPar1: A parallel implementation of SPSPL, with the Eval Monad, first attempt.
- SPSPLPar2: More optimized parallel implementation of SPSPL, with the Eval Monad.
- SPSPLPar3: Only evaluate in parallel till certain depth, then fall back to scanl1.
- SPSUBVecPLPar: A variation of SPSPLPar3 using Unboxed Vectors.

These are described further below.

2.3.1.1 SPSPL

Powerlist allow a simple prefix sum function:

```
sps \ \langle x \rangle = \langle x \rangle sps \ L = (sps \ u) \bowtie (sps \ v) where \ u \bowtie v = L^* \oplus L
```

which translates beautifully into haskell code:

```
import qualified Powerlist as P

sps :: Num a => (a -> a -> a) -> P.PowerList a -> P.PowerList a

sps _ [] = []

sps _ [x] = [x]

sps op 1 = P.zip (sps op u) (sps op v)

where (u, v) = P.unzip $ P.zipWith op (P.rsh 0 1) 1
```



This sequential implementation is presented to show the usefulness of powerlists and to benchmark parallel algorithms. The powerlist implementation used for this and other algorithms is backed by the *List* data structure of haskell, for its simplicity and flexibility.

2.3.1.2 SPSPLPar1

This was the first attempt at parallelizing SPSPL using the Eval monad. The algorithm is naturally recursive and divides the input into 2 equal sized arrays on which the SPSPLPar1 can be called recursively.

```
parSps1 :: (NFData a, Num a) => (a -> a -> a) -> P.PowerList a -> P.PowerList a

parSps1 _ [] = []

parSps1 _ [x] = [x]

parSps1 op l =

runEval

(do (u, v) <- rseq $ P.unzip $ P.zipWith op (P.rsh 0 l) l

u' <- rparWith rdeepseq (parSps1 op u)

v' <- rparWith rdeepseq (parSps1 op v)

rseq $ P.zip u' v')
```

The list implementation of powerlist has been used here. There is some bottleneck when we try to deconstruct the list into 2 halves, where several computations are being done linearly. Other problem is that we are creating a lot of intermediate lists, which require GC.

2.3.1.3 SPSPLPar2

In this variation, further optimization is done by using below techniques:

- We can parallelize the *unzip* operation that corresponds to deconstructing the list by observing that it is just creating 2 lists by elements from odd and even places in the input list. This is achieved by 2 simple functions *odds* and *evens*.
- We can parallelize zipWith operation that corresponds to \oplus operator by breaking the input lists into chunks and calling zipWith on the corresponding chunks of the 2 input lists and concatenating the output from each such call.
- There is a clever observation that since after right shift with 0 we are trying to run a zipWith operation, we can simply prepend 0 to the list and run the zipWith, since zipWith will automatically only run the operation on elements at the same position in both lists and ignore the extra last element in the list where 0 was added. This results in elimination of right shift operation.

```
odds :: [a] -> [a]
odds [] = []
odds [x] = [x]
odds (x:_:xs) = x : odds xs

evens :: [a] -> [a]
evens [] = []
evens [_] = []
evens (_:y:xs) = y : evens xs
```

2.3.1.4 SPSPLPar3

This version further improves the runtime by recursing only till a certain depth, thereby reducing the total number of sparks generated. We revert to scanl1 function for input arrays smaller than 2^d length, where d is the depth parameter, which is set to 5.



2.3.1.5 SPSUBVecPLPar

This algorithm is another implementation of **SPSPLPar3** but uses powerlist implementation using Unboxed Vectors.

```
import qualified UBVecPowerlist as UVP
   parSpsUBVec ::
        (NFData a, UV.Unbox a, Num a)
     => (a -> a -> a)
     -> Int
     -> Int
     -> UVP.PowerList a
     -> UVP.PowerList a
   parSpsUBVec _ _ _ 1
10
     | UV.length 1 <= 1 = 1
   parSpsUBVec op cs d l
     | d > 4 =
13
       runEval
         (do k <- rseq $ UVP.shiftAdd 1</pre>
             u <- rpar (UVP.filterOdd k)
16
             v <- rpar (UVP.filterEven k)</pre>
17
18
             u' <- rparWith rdeepseq (parSpsUBVec op cs (d - 1) u)
             v' <- rparWith rdeepseq (parSpsUBVec op cs (d - 1) v)
             UVP.parZip (rparWith rdeepseq) cs u' v')
22
   parSpsUBVec op _ _ 1 = UV.scanl1 op 1
```

This is expected to be faster because:

- Unboxed Vectors are more memory friendly.
- We introduce some additional operations like *shiftAdd* and *filterUsing* which directly execute the shift and add operation using mutable vectors.

2.3.2 Ladner Fischer

Another algorithm due to Ladner and Fischer can be implemented using powerlist as follows:

$$ldf \langle x \rangle = \langle x \rangle$$
$$ldf(p \bowtie q) = (t^* \oplus p) \bowtie t$$
$$where \ t = ldf(p \oplus q)$$

And here is the equivalent sequential implementation in haskell:



Again since the algorithm is naturally recursive over half the input array elements, it can be parallelized using the techniques used for SPS.

2.3.2.1 LDFPar

This is the parallel implementation of LDF algorithm, using the eval monad.

```
parLdf ::
        NFData a
2
     => Num a =>
          (a -> a -> a) -> Int -> Int -> P.PowerList a -> P.PowerList a
   parLdf _ _ _ [] = []
   parLdf _ _ _ [x] = [x]
6
   parLdf op cs d l
     | d > 4 =
       runEval
         (do p <- rpar (odds 1)</pre>
10
            q <- rpar (evens 1)
             _ <- rseq p
             _ <- rseq q
13
            pq <- rseq (P.parZipWith rdeepseq cs op p q)
14
            t <- rparWith rdeepseq (parLdf op cs (d - 1) pq)
            k <- rseq (P.parZipWith rdeepseq cs op (0 : t) p)
            rseq $ P.zip k t)
17
   parLdf op _ _ 1 = sequentialSPS op 1
```

We can again use all the previous improvements introduced in parallel versions of SPS algorithms (SPSPLPar1, SPSPLPar2, SPSPLPar3) to further optimize parLdf.

2.3.2.2 LDFUBVecPLPar

This algorithm is another implementation of LDFPar. As with **SPSUBVecPLPar** this algorithm uses Unboxed Vector as the powerlist implementation.

```
import qualified UBVecPowerlist as UVP
2
   parLdfUBVec ::
        (NFData a, UV.Unbox a, Num a)
     => (a -> a -> a)
     -> Int
     -> <u>Int</u>
     -> UVP.PowerList a
     -> UVP.PowerList a
   parLdfUBVec _ _ _ 1
10
     | UV.length 1 <= 1 = 1
   parLdfUBVec op cs d l
     | d > 4 =
       runEval
14
         (do p <- rpar $ UVP.filterOdd 1</pre>
15
             q <- rpar $ UVP.filterEven 1
             _ <- rseq p
17
             _ <- rseq q
18
             pq <- UVP.parZipWith (rparWith rdeepseq) op cs p q
19
             t <- rpar (parLdfUBVec op cs (d - 1) pq)
20
             k <- rseq $ UVP.shiftAdd2 t p</pre>
             UVP.parZip (rparWith rdeepseq) cs k t)
   parLdfUBVec op _ _ 1 = UV.scanl1 op 1
```




It has the same advantages as the SPSUBVecPLPar algorithm above. Note the use of shiftAdd2 and filterOdd and filterEven functions that use mutable vectors and hence might consume less memory.

2.3.2.3 LDFChunkUBVecPLPar

This algorithm uses another flavor of **LDFUBVecPLPar** where we first split the input into chunks, then run **LDFUBVecPLPar** over each of these chunks and then combine the results using a technique due to Bleloch [2].

The diagram below shows how the split and combine works. Basically we divide the input into chunks and call our function on each of them in parallel. Then we store the total sum of each chunk (last element) into another auxiliary chunk. We then scan this auxiliary chunk generating an array of chunk increments that are added to all the elements in their respective chunks.



Advantages of this technique are as follows.

- Since we use Unboxed Vector, splitting into chunks takes time proportional to the number of chunks.
- Chunk size controls parallelism of the algorithm, making it more scalable than previous implementations.

Here is the implementation:

```
parLdfChunkUBVec ::
    (NFData a, UV.Unbox a, Num a)

=> (a -> a -> a)

-> Int

-> UVP.PowerList a

-> UVP.PowerList a

parLdfChunkUBVec _ _ 1
```



```
| UV.length 1 <= 1 = 1
   parLdfChunkUBVec ops cs 1 = runEval $ parLdfChunkVec' ops chunks
     where
       n = UV.length 1
       chunkSize = 2 ^ cs
       chunks = S.chunksOf chunkSize 1
       parLdfChunkVec' ::
            (NFData a, UV.Unbox a, Num a)
         => (a -> a -> a)
         -> [UVP.PowerList a]
         -> Eval (UVP.PowerList a)
       parLdfChunkVec' _ [] = return UV.empty
19
       parLdfChunkVec' op vChunks = do
20
         resChunks <- parList rseq (parLdfUBVecNC op cs <$> vChunks)
21
         res <- rseq $ UV.concat resChunks
               -- Get last element of each block
         lastelems <- parList rdeepseq (UV.last <$> resChunks)
         lastScan <- rseq (UV.fromList $ sequentialSPS op lastelems)</pre>
25
         rseq $
26
           UV.create $ do
27
             m <- UV.thaw res
28
                   -- Not sure how to parallelise here!
29
            mergeChunks (n - 1) (UV.tail $ UV.reverse lastScan) m
             return m
       mergeChunks i lastScan m
         | i > chunkSize = do
           let ad = UV.head lastScan
           go m chunkSize i ad 0
35
           mergeChunks (i - chunkSize) (UV.tail lastScan) m
36
         | otherwise = return ()
       go m chs start v i
         | i < chs = do
39
           curr <- M.unsafeRead m (start - i)</pre>
40
           M.unsafeWrite m (start - i) (curr + v)
           go m chs start v (i + 1)
42
         | otherwise = return ()
43
```

Here **parLdfUBVecNC** is the implementation of **LDFUBVecPLPar** without parallelizing (using chunks) the secondary operators like *zipWith* since we already break the input into chunks at the start. I was unable to implement the addition of increments from auxiliary chunk to corresponding chunks in parallel using a mutable vector. Hence this implementation is still not optimal.

2.3.3 Batcher Merge Sort

This is another application of powerlist where a simple sorting scheme is given by:

$$sort \langle x \rangle = \langle x \rangle$$

 $sort(p \bowtie q) = (sort p) merge (sort q)$

We could use any merge function here to merge the 2 sorted sub-lists. The Batcher scheme [1] to merge 2 sorted lists can be expressed in terms of powerlist as the below infix operator bm



Here, a comparison operator \updownarrow has been used which is implemented as the minMaxZip function in haskell code. The operator is applied to a pair of equal length powerlists, p, q, and it creates a single powerlist by setting the $2i^{th}$ element to p_i min q_i and setting the $(2i+1)^{th}$ element to p_i max q_i , where p_i and q_i are the i^{th} elements of each of the 2 lists.

Here is the minMaxZip function for powerlist of vectors

```
minMaxZip :: (V.Unbox a, Ord a) => PowerList a -> PowerList a -> PowerList a
   minMaxZip xs ys =
     V.create $ do
       m <- M.new n
       write m 0
       return m
6
     where
       n = V.length xs + V.length ys
       write mv i
         | i < n = do
          let p = xs V.! (i 'div' 2)
          let q = ys V.! (i 'div' 2)
12
          M.unsafeWrite mv i (p 'min' q)
          M.unsafeWrite mv (i + 1) (p 'max' q)
14
          write mv (i + 2)
         | otherwise = return ()
16
```

This gives us the following batcher merge sort implementation in haskell

```
batcherMergeSort :: (Ord a, V.Unbox a) => P.PowerList a -> P.PowerList a
   batcherMergeSort 1
     | V.length 1 <= 1 = 1
   batcherMergeSort 1 = sortp 'batcherMerge' sortq
     where
       sortp = batcherMergeSort p
       sortq = batcherMergeSort q
       p = P.filterOdd 1
       q = P.filterEven 1
10
   batcherMerge ::
        (Ord a, V.Unbox a) => P.PowerList a -> P.PowerList a -> P.PowerList a
12
     | V.length x == 1 = V.fromList [hx 'min' hy, hx 'max' hy]
14
     where
15
       hx = V.head x
       hy = V.head y
17
   batcherMerge x y = P.minMaxZip rv su
18
19
       rv = r 'batcherMerge' v
20
       su = s 'batcherMerge' u
21
       r = P.filterOdd x
22
       v = P.filterEven y
```



```
s = P.filterEven x
u = P.filterOdd y
```

We use all the techniques used in the previous scan algorithms to come up with this parallel sort algorithm:

```
parBatcherMergeSort ::
        (NFData a, Ord a, V.Unbox a) => Int -> P.PowerList a -> P.PowerList a
   parBatcherMergeSort _ 1
     | V.length 1 <= 1 = 1
   parBatcherMergeSort d 1
     | d > 10 =
6
       runEval
         (do p <- rpar $ P.filterOdd 1</pre>
            q <- rpar $ P.filterEven 1
             _ <- rseq p
             sortp <- rparWith rdeepseq (parBatcherMergeSort (d - 1) p)
12
             sortq <- rparWith rdeepseq (parBatcherMergeSort (d - 1) q)</pre>
             parBatcherMerge d sortp sortq)
14
   parBatcherMergeSort _ l = V.fromList $ defaultSort $ V.toList l
16
   parBatcherMerge ::
17
        (Ord a, V.Unbox a)
     => Int
19
     -> P.PowerList a
20
     -> P.PowerList a
21
     -> Eval (P.PowerList a)
   parBatcherMerge d x y
     | d > 10 = do
       r <- rseq $ P.filterOdd x
       v <- rseq $ P.filterEven y
26
       rv <- parBatcherMerge (d - 1) r v
27
       s <- rseq $ P.filterEven x
28
       u <- rseq $ P.filterOdd y
       su <- parBatcherMerge (d - 1) s u
30
       rparWith rdeepseq $ P.minMaxZip rv su
   parBatcherMerge _ x y = rseq (merge x y)
```

The *merge* function call at line 32 is the sequential *merge* of mergesort algorithm implemented using mutable vectors. Again this is used to reduce the number of spark generated, as this algorithm is already highly parallel.

3 Benchmark

The benchmark results of various algorithms are listed in this section. Various combinations of chunk sizes and input size were tried, together with threadscope analysis of individual runs.

3.1 Setup

All benchmarks were performed on an 8 core Intel i9-9900K CPU @ 3.60 GHZ (32G) running Debian 11 (Bullseye).



3.2 Results

The summary results for all the algorithms is listed first, followed by the details about the best ones. Detailed results for each of the algorithm can be viewed on github here.

All results listed below are for arrays / lists of input length 2^{20} .

3.2.1 Scan

Following table summarizes the results for SPS algorithm variations:

Algo Name	Num Cores	${\bf Chunk Size}$	Runtime (ms)	Improvement
SPSPL	1	-	5232	-
SPSPLPar1	8	-	1506	3.47X
SPSPLPar2	8	256	1483	3.52X
SPSPLPar3	8	512	1397	3.74X
SPSUBVecPLPar	8	1024	520.3	10.05X

Following table summarizes the results for LDF algorithm variations:


Algo Name	Num Cores	ChunkSize	Runtime (ms)	Improvement
LDF	1	_	490.7	-
LDFPar	8	512	392.1	1.25X
LDFUBVecPLPar	8	1024	171.4	2.86X
LDFChunkUBVecPLPar	8	2^{10}	97.94	5.03X

We discuss the results of SPSUBVecPLPar, LDFUBVecPLPar and LDFChunkUBVecPLPar in further detail.

SPSUBVecPLPar


SPSUBVecPLPar is a significant improvement over SPS and also the list based parallel implementations of SPS, that is SPSPLPar1, SPSPLPar2 and SPSPLPar3. This can be attributed to using much less memory due to the use of Unboxed Vectors and mutable vectors for helper functions.

We experiment with many different chunk sizes, and 1024 performs best, as can be seen in the below graph from criterion html output:



Here are the threadscope results, as we can see there are too many sparks generated:






${\bf LDFUBVecPLPar}$

LDFUBVecPLPar performs even better, since the algorithm itself has better run time. As before, here is the variation with different chunk sizes, again 1024 works best:





Here are the threadscope results, load looks well balanced:






${\bf LDFChunkUBVecPLPar}$

LDFChunkUBVecPLPar performs the best with large enough chunk size, as we split the input from the top. Note that here chunksize is equal to 2^x where x is the number shown in the graph below:



Here are the threadscope results:






Due to the linear implementation of processing auxiliary chunk, we see parallelism limited to part of the run. We might be able to further improve the run time, if the last phase is implemented in parallel over single mutable vector.

3.2.2 Sort

Algo Name	Num Cores	Runtime (ms)	Improvement				
BATCHER	1	3929	-				
BATCHER	8	1721	2.28X				



The batcher sort, even though being a highly parallel algorithm, does not perform that well compared to a more traditional sort like quicksort. The obvious reason I could think of was because of all the copying and merging of intermediate arrays that is needed during the merge phase. Here is the threadscope analysis that shows how well batcher sort parallelizes:



4 Conclusion and Future Work

To summarize, we can say that powerlist provides a new abstraction to come up with recursive and parallel algorithms for several different use cases. The algorithms using powerlist parallelize very well, as can be seen



from threadscope captures. It is a challenge though to scale these parallel algorithms since they are recursive in nature, which inherently requires splitting and merging of input, thereby needing more memory. Simple iterative implementations of scan are difficult to beat with such algorithms.

We have several possibilities to extend this work.

- Explore other powerlist application algorithms like FFT.
- Exploit the commutative laws of scalar functions over powerlist operators to further parallelize the implemented algorithms.
- Try to use parallel libraries like massive that support nested parallelism.
- Experiment with RTS GC settings.

References

- [1] J. Misra, "Powerlist: A structure for parallel recursion," ACM Trans. Program. Lang. Syst., vol. 16, p. 1737–1767, nov 1994.
- [2] G. E. Blelloch, "Prefix sums and their applications," 1990.